

12. 电力变压器铁心柱截面的优化设计—1stOpt 整数规划案例分析

整数规划是运筹学重要组成部分，电力变压器铁心柱截面的优化设计既是一个典型的整数规划问题，也曾是全国大学生电工数学建模竞赛题。在此演示如何用 1stOpt 求解处理该类问题。

12.1 问题描述

为了充分利用线圈内空间又便于生产管理，心式铁心柱截面常采用多级阶梯型结构，如图 12.1 示三级结构。截面在园内上下及左右轴均对称。阶梯型的每级都是由多层相同宽度的硅钢片叠加而成。由于制造工艺的要求，不同级硅钢片的宽度要求为 5 的倍数（单位：毫米），每级宽度差值也必须是大于等于 5，此外还要求最小硅钢片的宽度不小于某阈值。




图 12.1. 三级铁心柱截面结构示意图

试求当铁心柱外接圆直径 $d = 200$ (mm)，宽度阈值 $W_{\min} = 30$ (mm)，每级硅钢片单片厚度分别用 h_1 、 h_2 和 h_3 表示，在此先取相同的值，即 $h_1 = h_2 = h_3 = 0.35$

(mm)，确定此时每级硅钢片的宽度 W 及厚度 T ，使得园横断面积内硅钢片叠加面积之和最大。

12.2 模型建立

参照图 12-1，很容易建立如下求最大值优化模型目标函数：

$$\text{Max. } A = 2 \cdot (W_1 \cdot T_1 + W_2 \cdot T_2 + W_3 \cdot T_3) \quad (12-1)$$

其中 W_1 、 W_2 和 W_3 分别代表三种规格硅钢片的宽度， T_1 、 T_2 及 T_3 分别是半截面园三级硅钢片的厚度，由此可计算出全横截面内矩形面积之和，也即目标函数如上式(12-1)。

厚度 T_1 、 T_2 、 T_3 是由各级单片硅钢片叠加，因为各级硅钢片单片厚度已知，均为 0.35mm，由此厚度问题就转换成每级各用了多少片硅钢片？对应半截面，分别设为 P_1 、 P_2 和 P_3 ，则目标函数由式(12-1)变为下式(12-2)：

$$\text{Max. } A = 2 \cdot (W_1 \cdot P_1 \cdot h_1 + W_2 \cdot P_2 \cdot h_2 + W_3 \cdot P_3 \cdot h_3) \quad (12-2)$$

下一个需要考虑的问题是第一级硅钢片上下对称叠加模式问题，如图 12-2 所示，有两种模式，一种是以水平中轴线为分界线，上下完全对称，上下硅钢片数之和一定为偶数，称之为偶数对称方案；另一种是水平中轴线刚好经过一片第一级硅钢片的中间，其余上下对称，显然这种模式下的第一级硅钢片总数一定为奇数，也即奇数对称方案。采用不同的对称方案，结果也会不同。




图 12.2. 第一级硅钢片上下对称模式：偶数对称与奇数对称
约束条件：

$$\text{S.T.} \begin{cases} W_{\min} \leq W_i \leq d & i = 1 \dots 3 & (a) \\ Z_i \leq \frac{d}{2} & i = 1 \dots 3 & (b) \\ W_i \bmod 5 = 0 & i = 1..3 & (c) \\ W_i \geq W_{i+1} + 5 & i = 1..2 & (d) \end{cases} \quad (12-3)$$

上诉约束条件中：

- a) 每级宽度 W 小于截面圆直径 d ，同时大于规定的最小值阈值 $W_{\min}$ ；
- b) 图 1 中所示 Z_1 、 Z_2 和 Z_3 小于半径 $(d/2)$ ，确保各级硅钢片均在截面圆内；
- c) “mod”是求余函数，确保每级宽度 W 为 5 的倍数；
- d) 要求每级宽度不同，因为每级的宽度为 5 的倍数，因此有 $W_1 \geq W_2 + 5$, $W_2 \geq W_3 + 5$ 。

参考图 12-1，约束条件中 Z_1 、 Z_2 和 Z_3 的计算公式如下：

偶数对称方案：

$$Z_1^2 = \left(\frac{W_1}{2}\right)^2 + (P_1 \cdot h_1)^2 \quad (12-4)$$

$$Z_2^2 = \left(\frac{W_2}{2}\right)^2 + (P_1 \cdot h_1 + P_2 \cdot h_2)^2 \quad (12-5)$$

$$Z_3^2 = \left(\frac{W_3}{2}\right)^2 + (P_1 \cdot h_1 + P_2 \cdot h_2 + P_3 \cdot h_3)^2 \quad (12-6)$$

奇数对称方案：

$$Z_1^2 = \left(\frac{W_1}{2}\right)^2 + \left(P_1 \cdot h_1 + \frac{h_1}{2}\right)^2 \quad (12-7)$$

$$Z_2^2 = \left(\frac{W_2}{2}\right)^2 + \left(P_1 \cdot h_1 + P_2 \cdot h_2 + \frac{h_1}{2}\right)^2 \quad (12-8)$$

$$Z_3^2 = \left(\frac{W_3}{2}\right)^2 + \left(P_1 \cdot h_1 + P_2 \cdot h_2 + P_3 \cdot h_3 + \frac{h_1}{2}\right)^2 \quad (12-9)$$

12.3 1stOpt 求解

基于前面分析，编写如下 1stOpt 求解代码。

偶数方案代码 12-1

```

Algorithm = DE1;
Constant d=200, Wmin=30;
Constant H(3)=[0.35,0.35,0.35];
Constant Hmin=MinConst(H);
IntParameter P(3)=[0,d/Hmin/2], W(3)=[Wmin,d];

```

```

MaxFunction (W1*P1*H1+W2*P2*H2+W3*P3*H3)*2;
    W1 mod 5 = 0;
    W2 mod 5 = 0;
    W3 mod 5 = 0;
    W3+5<=W2;
    W2+5<=W1;
    (d/2)^2>=(W1/2)^2+(P1*H1)^2;
    (d/2)^2>=(W2/2)^2+(P1*H1+P2*H2)^2;
    (d/2)^2>=(W3/2)^2+(P1*H1+P2*H2+P3*H3)^2;

```

奇数方案代码 12-2

```

Algorithm = DE1;
Constant d=200, Wmin=30;
Constant H(3)=[0.35,0.35,0.35];
Constant Hmin=MinConst(H);
IntParameter P(3)=[0,d/Hmin/2], W(3)=[Wmin,d];
MaxFunction (W1*P1*H1+W2*P2*H2+W3*P3*H3)*2+W1*H1;
    W1 mod 5 = 0;
    W2 mod 5 = 0;
    W3 mod 5 = 0;
    W3+5<=W2;
    W2+5<=W1;
    (d/2)^2>=(W1/2)^2+(P1*H1+H1/2)^2;
    (d/2)^2>=(W2/2)^2+(P1*H1+P2*H2+H1/2)^2;
    (d/2)^2>=(W3/2)^2+(P1*H1+P2*H2+P3*H3+H1/2)^2;

```

上述代码中，“Algorithm = DE1;”：缺省的 UGO 算法对整数规划问题效果不好，在此使用差分进化算法；“Constant Hmin=MinConst(H);”：Hmin 是取三级硅钢片单片厚度值最小者，案例中硅钢片单片厚度均为 0.35，因此 Hmin 也为 0.35，如果各级硅钢片单片厚度不同，会自动取最小值；“IntParameter P(3)=[0,d/Hmin/2], W(3)=[Wmin,d];”：定义整数参数 P1、P2 和 P3，各级硅钢片总数的一半（对应半截面园），其值范围在[0, d/Hmin/2]之间，d/Hmin/2 表示半园内可能最大的硅钢片数，各级硅钢片的宽度 W1、W2 和 W3 应在最小设定宽度及园直径之间。

偶数方案与奇数方案主要有两个不同处：一是奇数方案目标函数多了“W1*H1”部分，实际上是过水平中心线硅钢片的面积，二是计算 Z 时公式略有不同，多了“H1/2”部分。

使用“For”和“Sum”语句，上述代码可写成如下更为简练通用的形式：

偶数方案代码 12-3

```

Algorithm = DE1;
Constant n=3, d=200, Wmin=30;
Constant H(n)=[0.35,0.35,0.35];
Constant Hmin=MinConst(H);

```

```

IntParameter P(n)=[0,d/Hmin/2], W(n)=[Wmin,d];
PassParameter Area=pi*(d/2)^2, Length=Sum(i=1:n,P,H)(P*H)*2;
MaxFunction Sum(i=1:n,W,P,H)(W*P*H)*2;
    For(i=1:n,W)(W mod 5 = 0);
    For(i=1:n-1)(W[i+1]+5<=W[i]);
    For(i=1:n)((d/2)^2>=(W[i]/2)^2+(Sum(j=1:i)(P[j]*H[j]))^2);

```

奇数方案代码 12-4

```

Algorithm = DE1;
Constant n=3, d=200, Wmin=30;
Constant H(n)=[0.35,0.35,0.35];
Constant Hmin=MinConst(H);
IntParameter P(n)=[0,d/Hmin/2], W(n)=[Wmin,d];
PassParameter Area=pi*(d/2)^2, Length=Sum(i=1:n,P,H)(P*H)*2+H1;
MaxFunction Sum(i=1:n,W,P,H)(W*P*H)*2+W1*H1;
    For(i=1:n,W)(W mod 5 = 0);
    For(i=1:n-1)(W[i+1]+5<=W[i]);
    For(i=1:n)((d/2)^2>=(W[i]/2)^2+(Sum(j=1:i)(P[j]*H[j])+H1/2)^2);

```

上述代码中“n”表示分级数；“PassParameter”一句表示计算结束后会输出圆的面积及所有硅钢片相加厚度，该厚度实际上应该小于直径d。

运行上述代码，可得结果如下：

偶数方案	奇数方案
目标函数值(最大): 26677	目标函数值(最大): 26701.5
p1: 124	p1: 124
p2: 80	p2: 79
p3: 54	p3: 55
w1: 180	w1: 180
w2: 140	w2: 140
w3: 85	w3: 85
传递参数(PassParameter):	传递参数(PassParameter):
area: 31415.9265358979	area: 31415.9265358979
length: 180.6	length: 180.95

从上述结果可看出：

- 1) 偶数方案：最大叠加面积 26677，三级硅钢片各自宽度分别为 180、140 和 85，对应硅钢片数分别为：124*2、80*2 和 54*2；
- 2) 奇数方案：最大叠加面积 26701，三级硅钢片各自宽度分别为 180、140 和 85，对应硅钢片数分别为：124*2+1、79*2 和 55*2；

如果仅从目标函数值看，奇数方案更好。

12.4 通用模型代码

前面代码按方案不同分为偶数和奇数两种，下面引入识别符号 k，k 为 0 时表示偶数方案，k 为 1 时奇数方案，代码如下：

通用代码 12-5

```
Algorithm = DE1;
Constant k=0; //k=0: 偶数对称; k=1: 奇数对称
Constant n=3, d=200, Wmin=30;
Constant H(n)=[0.35,0.35,0.35];
Constant Hmin=MinConst(H);
IntParameter P(n)=[0,d/Hmin/2], W(n)=[Wmin,d];
PassParameter Area=pi*(d/2)^2, Length=Sum(i=1:n,P,H)(P*H)*2+H1*k;
MaxFunction Sum(i=1:n,W,P,H)(W*P*H)*2+W1*H1*k;
    For(i=1:n,W)(W mod 5 = 0);
    For(i=1:n-1)(W[i+1]+5<=W[i]);
    For(i=1:n)((d/2)^2>=(W[i]/2)^2+(Sum(j=1:i)(P[j]*H[j])+k*H1/2)^2);
```

仅需改变上述代码中的第 2,3,4 句中的具体值，即可处理不同情境下的电力变压器铁心柱截面的优化设计。

1) 情境一： n=4， d=150， Wmin=30, H(n)=[0.35,0.35,0.35,0.35]

偶数方案	奇数方案
目标函数值(最大): 15610	目标函数值(最大): 15603
p1: 76	p1: 76
p2: 52	p2: 61
p3: 43	p3: 39
p4: 31	p4: 25
w1: 140	w1: 140
w2: 120	w2: 115
w3: 90	w3: 85
w4: 50	w4: 50
传递参数(PassParameter):	传递参数(PassParameter):
area: 17671.4586764426	area: 17671.4586764426
length: 141.4	length: 141.05

此情景中偶数方案为好

2) 情境二： n=4， d=250， Wmin=30, H(n)=[0.45,0.40,0.35,0.3]

偶数方案	奇数方案
目标函数值(最大): 43403.5	目标函数值(最大): 43425.75
p1: 108	p1: 94
p2: 74	p2: 81
p3: 67	p3: 71
p4: 53	p4: 56
w1: 230	w1: 235

w2: 195 w3: 145 w4: 85 传递参数(PassParameter): area: 49087.3852123405 length: 235.1	w2: 200 w3: 150 w4: 90 传递参数(PassParameter): area: 49087.3852123405 length: 233.15
---	--

此情景中奇数方案为好

3) 情境三：n=4，d=250，Wmin=30，H(n)=[0.3,0.35,0.4,0.45]

偶数方案	奇数方案
目标函数值(最大): 43445.5 p1: 142 p2: 92 p3: 63 p4: 39 w1: 235 w2: 200 w3: 150 w4: 85 传递参数(PassParameter): area: 49087.3852123405 length: 235.1	目标函数值(最大): 43421 p1: 141 p2: 93 p3: 67 p4: 35 w1: 235 w2: 200 w3: 145 w4: 85 传递参数(PassParameter): area: 49087.3852123405 length: 235.1

该情境与与情境二只是每级硅钢片厚度数据不同，但结果迥异，是偶数方案为好。

同一情境下，奇偶数方案孰优孰劣取决于具体的数据。计算时最好多运行几次，看结果是否稳定唯一，去满足条件的最好结果。

12.5 结论

对于电力变压器铁心柱截面的优化设计整数规划问题，所提出的 1stOpt 通用模型及程序代码，可以对两种模式（偶数及奇数方案）及各种规格条件下进行优化求解。